


AD-A099 492 ARMY MISSILE COMMAND REDSTONE ARSENAL AL TEST AND EVA--ETC F/6 9/3
AN AUTOMATED MIL-STD 461 EMISSIONS TEST PROCEDURE. (U)
JUL 80 R A SNEAD, L G STOUDER/NIRE


UNCLASSIFIED DRSMI/RT-80-23-TR

SBIE-AD-E950 128

ML



END
DATA
TELETYPE
6 81
DTIC



MICROCOPY RESOLUTION TEST CHART
NATIONAL BUREAU OF STANDARDS 1963 A

AD A 099 492

(2)

LEVEL III

AD-E 950128

TECHNICAL REPORT RT-80-23

AN AUTOMATED MIL-STD 461 EMISSIONS TEST PROCEDURE

R. A. Snead and L. G. Stoudermire
Test and Evaluation Directorate
US Army Missile Laboratory

22 July 1980

DTIC
ELECTE
S JUN 1 1981 D
B

Approved for public release; distribution unlimited.



U.S. ARMY MISSILE COMMAND

Redstone Arsenal, Alabama 35809

DMC FILE COPY

DISPOSITION INSTRUCTIONS

**DESTROY THIS REPORT WHEN IT IS NO LONGER NEEDED. DO NOT
RETURN IT TO THE ORIGINATOR.**

DISCLAIMER

**THE FINDINGS IN THIS REPORT ARE NOT TO BE CONSTRUED AS AN
OFFICIAL DEPARTMENT OF THE ARMY POSITION UNLESS SO DESIG-
NATED BY OTHER AUTHORIZED DOCUMENTS.**

TRADE NAMES

**USE OF TRADE NAMES OR MANUFACTURERS IN THIS REPORT DOES
NOT CONSTITUTE AN OFFICIAL INDORSEMENT OR APPROVAL OF
THE USE OF SUCH COMMERCIAL HARDWARE OR SOFTWARE.**

UNCLASSIFIED
SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

DD FORM 1 JAN 73 1473 EDITION OF 1 NOV 68 IS OBSOLETE

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

SECURITY CLASSIFICATION OF THIS PAGE(When Data Entered)

SECURITY CLASSIFICATION OF THIS PAGE(When Data Entered)

CONTENTS

	Page No.
I. INTRODUCTION	3
II. PROGRAM DESCRIPTION.	4
III. SUBROUTINE DESCRIPTIONS.	7
A. DEV.	7
B. ARRAY.	7
C. PLINIT	8
D. GRAT	8
E. TITLE1 and TITLE-OUT	10
F. ANT.	11
G. DRAW	11
H. SETUP.	12
I. FREQ	12
J. BWP.	13
K. SWEEP.	14
L. REDRL.	14
M. REDSA.	14
N. CHOP	15
O. SMOOTH	16
P. BCOR	17
Q. ANTCOR and ANTCORLIM	17
REFERENCES	20
APPENDIX A. RESOLUTION AND VIDEO BANDWIDTH SETTINGS FOR EACH FREQUENCY BAND	21
APPENDIX B. PROGRAM TO CALCULATE SEMILOG SLOPES AND INTERCEPTS.	22
APPENDIX C. LIST OF PROGRAM VARIABLE ASSIGNMENTS.	23
APPENDIX D. EQUIPMENT LIST.	24
APPENDIX E. WHY DID WE...?.	26
APPENDIX F. TYPICAL TEST RESULTS.	27

Accession For	
NTIS GRA&I <input checked="" type="checkbox"/>	
DTIC TAB <input type="checkbox"/>	
Unannounced <input type="checkbox"/>	
Justification _____	
By _____	
Distribution/ _____	
Availability Codes	
Dist	Avail and/or Special
A	

I. INTRODUCTION

Since its inception in the early 1960s, MIL-STD-461 and its associated notes have been one of the primary standards for evaluating the electromagnetic compatibility of military electrical and electronic equipment. The standard, and its associated standard (MIL-STD-462) outline applicable limits and test procedures to minimize electromagnetic interference (EMI) between systems and components.

These two standards require conducted and radiated emissions and susceptibility tests in a number of frequency bands from 30 Hz to 1.5 GHz (the tests required for emissions are shown in figure 1). These tests require a large number of instrument parameter and frequency effect corrections due to the wide frequency bands over which they are run. This requirement (repetitive type tests with large numbers of calculated corrections) indicated that an automated test system would significantly reduce the time required to run the test.

CONDUCTED EMISSIONS TESTS

A N AF

CE01 CE01	POWER LEADS DC POWER LEADS	30 Hz - 20 KHz 30 Hz - 50 KHz	X	X	X
CE02 CE02	CONTROL & SIGNAL LEADS AC POWER LEADS	30 Hz - 20 KHz 10 KHz - 50 KHz	X	X	X
CE03 CE03	POWER LEADS CONTROL & SIG LEADS	20 KHz - 50 MHz 30 Hz - 50 KHz	X	X	X
CE04 CE04	CONTROL & SIG LEADS POWER LEADS	20 KHz - 50 MHz 50 KHz - 50 MHz	X	X	X
CE05 CE06	INVERSE FILTER METH CONTROL & SIG LEADS	30 Hz - 50 MHz 50 KHz - 50 MHz	X	X	X
CE06 CE06	ANTENNA TERMINALS ANTENNA TERMINAL	10 KHz - 12.4 GHz 10 KHz - 12.4 GHz	X	X	X
CE07	POWER SOURCE-TAC.V	1.5 - 85 MHz	X		

RADIATED EMISSIONS TESTS

A N AF

RE01	MAGNETIC FIELD	30 Hz - 30 KHz	X	X	X
RE02 RE02:2.1	ELECTRIC FIELD	14 KHz - 10 GHz 14 KHz - 12.4 GHz	X	X	X
RE03	SPURIOUS & HARMONIC	10 KHz - 40 GHz	X	X	X
RE04	MAGNETIC FIELD	20 Hz - 50 KHz	X	X	X
RE05	ENG DRIVEN EQUIP	150 KHz - 1GHz	X	X	X
RE06 RE06	OVERHEAD POWER LINES OVERHEAD POWER LINES	150 KHz - 1 GHz 14 KHz - 1 GHz	X	X	X

A = ARMY VERSION

N = NAVY VERSION

AF = AIR FORCE VERSION

Figure 1. Emissions tests required by MIL-STD-461.

The automatic test system which we developed is based on an HP9825 computing controller and an HP8568A spectrum analyzer. This equipment was chosen for the following reasons:

- 1) The 9825 calculator system had been purchased previously and was available for use.
- 2) The requirements of interfacing the controller and spectrum analyzer are simplified using the HPIB. (Since the 8568A presents digitally processed data in stand-alone mode anyway, the A to D and D to A conversion was designed into the machine, not the HPIB interface.)
- 3) Since the 8568A spectrum analyzer functions are fully programmable from 100 Hz to 1.5 GHz, we needed to buy only one unit, rather than the three EMI receivers needed to run the same test, and operator intervention during the test was eliminated.

The actual equipment used in the test is listed in Appendix D. The optional equipment was used since we already had it, but only the antennas, plotter, spectrum analyzer and controller are necessary for running the test.

The algorithm used for acquiring and processing the data is a simple one consisting of the following steps:

- 1) Acquire the data.
- 2) Add the antenna correction factor.
- 3) Add the impulse bandwidth correction factor for broadband data.
- 4) Plot the data.

Further information concerning the program can be found in the program description section which follows.

The program outlined in the program listing has been run and checked against industry standard instrumentation (Singer EMI receivers) and calibrated impulse sources. The errors noted in these two comparisons have been limited to the tolerances of the receivers (less than 3 dB). (NOTE: This correlation assumes the EMI receivers are in peak mode.) Due to duty cycle effects, Singer average mode readings of signals will be significantly different from the 8568A readings for non-CW signals or non-50% duty cycle pulses.)

II. PROGRAM DESCRIPTION

As can be seen from the program listing and flow diagram (figure 2), this program is divided into two sections. The first one (lines 0 to 30) is a loop which controls the program flow by calling appropriate subroutines. This section also initializes the computational and data acquisition equipment.




Figure 2. RE02 and 2.1 program flow.

The second section (lines 31 to 378) contain the subroutines called by the control section.

Each of the subroutines will be explained in detail in the following sections, but some words of clarification are needed concerning the control section:

Line 3 - clears the HPIB interface

Line 9 - initializes the loop counter (r1). The number of times the loop is executed depends on the number of intervals into which the spectrum to be surveyed must be divided to prevent the appearance of the spectrum analyzer local oscillator skirts in the data.

Line 10 - initializes string variable A\$ which sets the broadband/narrowband mode (impulse bandwidth correction factors are not used during analysis of narrowband data). Note that the branch to BCOR, the impulse bandwidth correction subroutine, is only taken when broadband mode is set.

Line 23 - sets the narrowband mode at the end of the broadband measurement for each frequency band.

Line 24 - loop back to the start of the measurement routines so that the narrowband data collection can begin.

Line 26 - checks the loop counter (r1) to see if all measurement bands have been completed, looping back to line 10 if they have not.

Line 27 - stores the plotter pen.

```
0: "                      MIL SPEC 461 TEST PROGRAM":  
1: "                      SECTION RE02 (RADIATED EMISSIONS)":  
2: asb "DEV"  
3: cli 7  
4: gsb "PLINIT"  
5: gsb "ARRAY"  
6: gsb "orat"  
7: gsb "TITLE1"  
8: gsb "title-out"  
9: 1→r1  
10: "BB"→A$  
11: gsb "ANT"  
12: gsb "SETUP"  
13: gsb "FREQ"  
14: gsh "BWP"  
15: gsh "SWEEP"  
16: gsb "REDRL"  
17: gsb "REDSA"  
18: gsb "chop"  
19: if A$="BB";asb "smooth"  
20: if A$="BB";gsb "BCOR"
```

```

21: gsb "ANTCOK"
22: if A$=="NB";gto +3
23: "NB">A$
24: gto -12
25: i+r1>r1
26: if r1<7;gto -16
27: pent
28: dsp "END OF TEST";beep;end
29:

```

III. SUBROUTINE DESCRIPTIONS

A. DEV

Subroutine DEV uses the HPL device statement (dev) to associate string variables with peripheral logical unit numbers (HPIB addresses). Line 33 makes the following assignments:

<u>device</u>	<u>name</u>	<u>HPIB addr.</u>
HP8568 Spectrum Analyzer	sa	718
HP9866B Line Printer	lp	6
HP9872A Plotter	gr	705
HP9825 Controller	25	719
HP2631A Line Printer	ptr	714

The main reason for using the device statement is to facilitate program modification if the HPIB device address assignments must be changed. If peripheral devices were referenced directly by address, changing an HPIB address would mean modifying every statement which referenced that peripheral. By using the device statement this situation is averted, and only the device statement making the assignment must be changed if HPIB addresses are changed.

```

30: "***** SUBROUTINES *****";
31:
32: "DEV ASSIGNS THE HPIB ADDRESSES";
33: "DEV":dev "sa",718,"lp",6,"gr",705,"25",719,"ptr",714;ret
34:

```

B. ARRAY

Array dimensions the necessary storage arrays for the real and string variables. The HPL dim statement is equivalent to the FORTRAN DIMENSION statement.

There is one difference, however. In HPL the string variables must be dimensioned. The statement dim A\$[x,y] dimensions an array of x number of string variables, each of which is y characters long, assigned to string

array A\$. The statement dim A\$[x], dimensions a string variable, " characters long to be named A\$.

```
35: "ARRAY";
36: "ARRAY DIMENSIONS ALL ARRAYS";
37: dim K[100],Q[100],L[1001],A$[2],T$[64],D$[10],64]
38: ret
39:
```

C. PLINIT

This routine initializes (clears) the 9872A Graphics Plotter and sets the plotting window scale points P1 and P2. It also sets the minimum and maximum scale values in user units.

```
155: "PLINIT";
156: "INITIALIZES THE PLOTTER AND SETS THE VALUE OF P1 & P2";
157: wrt "gr", "IN"
158: wrt "gr", "IP 100,1000,13000,8000"
159: scl -100,2300,-60,180
160: ret
161:
```

D. GRAT

The purpose of this subroutine is to draw and label two semi-log plots (one broadband and one narrowband) on which the final data will be plotted. Variable Z contains an X offset (1200 for narrowband, 0 for broadband) used to distinguish between the plot positions for both drawing the graticles and plotting the data.

Lines 200 through 209 draw vertical logarithmic lines representing the frequencies of the points to be plotted. This requires a translation of the linear x coordinate to a log scale. The translation is accomplished by the following algorithm:

1. The points (0,0) and (amp-min, freq min) are placed at the same point mathematically. The x origin translation is accomplished by calculating an offset function:

$$\phi = \log f_{\min} - \text{offset}, \text{ or}$$

$$\log f_{\min} = \text{offset}.$$

2. Since the frequency range shown in the plot is 1000 plotter units wide, the number of points per integer log units is:

$$n = 1000 / (\log F_{\text{stop}} - \log F_{\text{start}}).$$

3. Using the relationships above, the x coordinate (frequency) of any point may be calculated by:

$$x = n (\log f - \text{offset}).$$

4. The amplitude (y coordinate) may be represented in a linear mode and in this subroutine the offsets are automatically calculated in the scl statement contained in PLINIT.

This algorithm is also used to calculate the points being plotted in the plotting portion of ANTCOR.

Lines 211 through 218 plot the eleven horizontal lines representing the linear dBuV scale (vertical coordinate lines). Lines 222 through 263 label the axes of the graph. Lines 264 through 274 draw and label the broadband and narrowband limits.

```
193: "qrat";
194: "draws the log graph on the plotter";
195: csiz 1,2,1,0
196: pen# 1
197: 0→Z
198: "nbloop":
199: plt -100+Z,-60,-2
200: iplt 0,240;iplt 1200,0;iplt 0,-240;iplt -1200,0
201: pen
202: plt 0+Z,-40,-2
203: for I=1.4e4 to 1e9
204: log(I)→D;2.060210915e2*(D-4.146128036)→H
205: plt H+Z,-40;plt H+Z,160,-1
206: I+tntint(D)→I
207: log(I)→D;2.060210915e2*(D-4.146128036)→H
208: plt H+Z,160;plt H+Z,-40,-1
209: I+tntint(D)→I
210: next I
211: for I=-40 to 160 by 40
212: iplt -1000,0,2
213: iplt 0,20,1
214: if I=160;gto +3
215: iplt 1000,0,2
216: iplt 0,20,1
217: pen
218: next I
219: if Z=1200;0→Z;gto +3
220: 1200→Z
221: gto "nbloop"
222: plt 0,165;lbl "RADIATED EMISSIONS (REO2 14kHz-1.0GHz) BB DATA"
223: plt 1200,165;lbl "RADIATED EMISSIONS (REO2.1 14kHz-1.0GHz) NB DATA"
224: csiz 1,2,1,90
225: plt 1025,-20;lbl "dBuV/m/MHz"
226: plt 2220,-20;lbl "dBuV/m"
227: csiz 1,2,1,0
228: plt -50,160;lbl "160"
229: plt -50,140;lbl "140"
230: plt -50,120;lbl "120"
231: plt -50,100;lbl "100"
232: plt -35,80;lbl "80"
233: plt -35,60;lbl "60"
234: plt -35,40;lbl "40"
```

```

235: plt -35,20;lbl "20"
236: plt -20,0;lbl "0"
237: plt -50,-20;lbl "-20"
238: plt -50,-40;lbl "-40"
239: plt -20,-45;lbl ".01"
240: plt 170,-45;lbl ".1"
241: plt 380,-45;lbl "1"
242: plt 580,-45;lbl "10"
243: plt 780,-45;lbl "100"
244: plt 980,-45;lbl "1000"
245: plt 300,-50;lbl "FREQUENCY IN MEGAHERTZ"
246: plt 1150,160;lbl "100"
247: plt 1150,140;lbl "140"
248: plt 1150,120;lbl "120"
249: plt 1150,100;lbl "100"
250: plt 1165,80;lbl "80"
251: plt 1165,60;lbl "60"
252: plt 1165,40;lbl "40"
253: plt 1165,20;lbl "20"
254: plt 1180,0;lbl "0"
255: plt 1150,-20;lbl "-20"
256: plt 1150,-40;lbl "-40"
257: plt 1180,-45;lbl ".01"
258: plt 1370,-45;lbl ".1"
259: plt 1580,-45;lbl "1"
260: plt 1780,-45;lbl "10"
261: plt 1980,-45;lbl "100"
262: plt 2180,-45;lbl "1000"
263: plt 1500,-50;lbl "FREQUENCY IN MEGAHERTZ"
264: "plot the spec limits":
265: plt 0,100,-2
266: plt 856,55
267: plt 1000,70,-1
268: plt 1200,34,-2
269: plt 1870,20
270: plt 2200,45,-1
271: plt 1010,72;lbl "BB"
272: plt 1010,67;lbl "LIMIT"
273: plt 2210,47;lbl "NB"
274: plt 2210,42;lbl "LIMIT"
275: ret
276:

```

E. TITLE1 and TITLE-OUT

TITLE1 inputs an array of string variables which serves as a graph data identification block. The input format is eight string variables, each 64 characters long. TITLE-OUT simply prints the strings input in TITLE1 on the line printer.

```

143: "TITLE1";
144: "INPUT THE DOCUMENTATION INFORMATION TO BE PLACED ON THE PLOTS";
145: ent "SYSTEM NAME?",T$;"SYSTEM NAME: "&T$→D$[1];""→T$;
146: ent "TEST DATE?",T$;"TEST DATE: "&T$→D$[2];""→T$;
147: "MICOM/DRSMI-RTR"→D$[3];
148: ent "YOUR NAME?",T$;"NAME: "&T$→D$[4];""→T$;
149: ent "TEST NUMBER?",T$;"TEST NUMBER: "&T$→D$[5];""→T$;
150: ent "MODE?",T$;"MODE: "&T$→D$[6];""→T$;
151: ent "POLARIZATION?",T$;"POLARIZATION: "&T$→D$[7];""→T$;
152: ent "TEST CONFIGURATION?",T$;"TEST CONFIGURATION: "&T$→D$[8];""→T$;
153: ret
154:
277: "title-out";
278: "outputs the annotation strings entered by the operator";
279: wrt "ptr",char(10)&char(13)
280: wrt "ptr",char(27)&"n"
281: for I=1 to 8
282: wrt "ptr",D$[I]
283: next I
284: wrt "ptr",char(27)&"o"
285: ret
286:

```

F. ANT

There are two objectives for this subroutine. The first is to prompt the user of the program to pick the correct antenna for the frequency band and change it when necessary. The second is to call the subroutine which draws the instrumentation configuration on the spectrum analyzer display screen. The program then stops and waits for the operator to verify the test setup.

```

87: "ANT";
88: "ANT PROMPTS THE USER TO PICK THE CORRECT ANTENNA FOR THE BAND";
89: if r1=1;gto +3
90: if r1=5;gto +5
91: if r1=6;gto +6
92: if r1>1 and r1<5;ret
93: dsp "ATTACH EMCO 3301 TO INPUT 1"
94: gto +4
95: dsp "ATTACH EMCO 3104 TO INPUT 1"
96: gto +2
97: dsp "ATTACH EMCO 3101 TO INPUT 1"
98: qsh "DRAW"
99: beep;stp
100: ret
101:

```

G. DRAW

This routine consists of a series of graphics commands written to the spectrum analyzer trace C memory. As a result of these commands, a suggested test setup is drawn on the screen of the spectrum analyzer.

```

162: "DRAW";
163: wrt "sa", "IP"
164: wrt "sa", "DTE"
165: wrt "sa", "A4,B4,KSM,KSo,D2"
166: wrt "sa", "PUPA 350,500"
167: wrt "sa", "PDPR -300,0,0,300,300,0,0,-300"
168: wrt "sa", "PUPR -250,25"
169: wrt "sa", "LB856BA@"
170: wrt "sa", "PUPA 300,525 PDPR 0,-100,100,0"
171: wrt "sa", "PR 0,10,20,0,0,-20,-20,0,0,10"
172: wrt "sa", "PUPR 20,10 PDPR 50,30 PUPR-50,-50 PDPR 50, 30"
173: wrt "sa", "PUPA 650,600 PDPR 300,0,0,-300,-300,0,0,300"
174: wrt "sa", "PUPR 50,-150 LBE.U.T.@"
175: wrt "sa", "PUPR -400,275 LBAPPROPRIATE ANTENNA@"
176: if r1=1;wrt "sa", "PUPR -300,-50 LBATTACH EMC0 3701 TO INPUT 1@"
177: if r1=5;wrt "sa", "PUPR -300,-50 LBATTACH EMC0 3104 TO INPUT 1@"
178: if r1=6;wrt "sa", "PUPR -300,-50 LBATTACH EMC0 3101 TO INPUT 1@"
179: wrt "sa", "PUPR -275,-10 PDPR-50,-175"
180: wrt "sa", "PUPA 350,175 LR1,0 METER SPACING@"
181: wrt "sa", "PUPR -125,25 PDPR 100,200"
182: wrt "sa", "PUPA 0,400 LR LEFT INPUT@ PDPR 125,125"
183: wrt "sa", "PUPA 200,900 LBRE02 INSTRUMENTATION SETUP@"
184: ret
185:

```

H. SETUP

This subroutine sets up the spectrum analyzer to collect the data using the following commands:

IP - Run the spectrum analyzer instrument preset routine. Set all controls and functions to their default states.

KSC - Set the display units to dB μ V.

AT0DB - Set the spectrum analyzer input attenuation to 0 db.

I1 - Select input 1. The HP8568A has two signal inputs. Input 1 is selected because it has a wider bandwidth (100 Hz - 1.5 GHz) than input 2 (100 KHz - 1.5 GHz).

```

40: "SETUP";
41: "SETS THE 8568A REFERENCE LEVEL";
42: wrt "sa", "IP KSC"
43: wrt "sa", "AT0DB I1"
44: ret
45:

```

I. FREQ

This subroutine passes the start and stop frequencies for the band to be displayed to the spectrum analyzer. The parameters passed are selected based on the current value of the loop counter (r1).

```

46: "FREQ";
47: "FREQ SELECTS THE FREQUENCY SPAN OF THE ANALYZER";
48: if r1=1;wrt "sa","FB.100MZ FA.014MZ";ret
49: if r1=2;wrt "sa","FB1.0MZ FA.1MZ";ret
50: if r1=3;wrt "sa","FB10.0MZ FA1.0MZ";ret
51: if r1=4;wrt "sa","FB20.0MZ FA10.0MZ";ret
52: if r1=5;wrt "sa","FB200.0MZ FA20.0MZ";ret
53: if r1=6;wrt "sa","FB1000.0MZ FA200.0MZ";ret
54: beep;wrt "lp","NUMBER OF VALID BANDS EXCEEDED IN FREQ";wrt "lp";end
55:

```

J. BWP

This subroutine passes the video and resolution bandwidth parameters to the spectrum analyzer, based on the current loop counter and the state of the BB/NB mode variable (A\$).

The resolution and video bandwidths used in this program represent an optimum set of parameters arrived at by balancing the following requirements:

- 1) The noise floor of the instrumentation system is required to be at least 6 dB below the MIL-STD 461 limits.
- 2) The sweep time of the spectrum analyzer display must be appropriate. Under some combinations of parameters, the sweep times may be as long as twenty-five minutes; much too long to run repetitive tests.
- 3) The difference in the narrowband and broadband measurement resolution bandwidth should be at least a factor of ten for all bands.
- 4) The video bandwidth should be as large as possible during the acquisition of broadband data in order to allow accurate spectral representation of fast transients present. For narrowband signals, it was determined that setting the resolution and video bandwidths equal produced the best correlation between the automated system and the EMI receivers generally used to run this test manually.

```

56: "BWP";
57: "BWP PASSES THE BANDWIDTH PARAMETERS TO THE 8568A";
58: if A$="BB";gto +3
59: if A$="NB";gto +10
60: "BROAD BAND PARAMETER SETUP";
61: if r1=1;wrt "sa","RB1.0KZ,VB3.0MZ"
62: if r1=2;wrt "sa","RB10.0KZ,VB3.0MZ"
63: if r1=3;wrt "sa","RB100.0KZ,VB3.0MZ"
64: if r1=4;wrt "sa","RB100.0KZ,VB3.0MZ"
65: if r1=5;wrt "sa","RB1.0MZ,VB3.0MZ"
66: if r1=6;wrt "sa","RB1.0MZ,VB3.0MZ"
67: gto +8
68: "NARROW BAND PARAMETER SETUP";
69: if r1=1;wrt "sa","RB100.0HZ,VB100.0HZ"
70: if r1=2;wrt "sa","RB300HZ,VB300HZ"
71: if r1=3;wrt "sa","RB3KZ,VB3KZ"
72: if r1=4;wrt "sa","RB3KZ,VB3KZ"
73: if r1=5;wrt "sa","RB3KZ,VB3KZ"
74: if r1=6;wrt "sa","RB10KZ,VB10KZ"
75: ret

```

K. SWEEP

This subroutine triggers the spectrum analyzer sweep, using the A trace memory (trace B is blanked). The trace is cleared and the max hold (peak) mode is set. If broadband mode is set, the spectrum analyzer is allowed to sweep continuously for approximately twenty seconds. This helps to pick up transient noise with a low duty cycle.

For narrowband measurements, the single sweep mode is set and the sweep is triggered. At the end of the sweep for both modes, the trace memory is set to "view" in order to prevent inadvertent changes in displayed data.

```
77: "SWEEP";
78: "SWEEP CAUSES THE 8568A TO SWEEP THE BAND & DISPLAY IT";
79: wrt "sa", "HD A1"
80: wrt "sa", "A2"
81: if A$="BB";wait 20000;gto +3
82: wrt "sa", "S2"
83: wrt "sa", "TS"
84: wrt "sa", "A3"
85: ret
86:
```

L. REDRL

The sole purpose of this subroutine is to read the current reference level from the spectrum analyzer and store it in variable R for later use.

```
186: "REDRL";
187: "READ THE 8568A REFERENCE LEVELS";
188: wrt "sa", "RL OA"
189: red "sa", R
190: wrt "sa", "HD"
191: ret
192:
```

M. REDSA

REDSA is the subroutine which reads the raw data stored in the spectrum analyzer trace memory into the 9825 controller for analysis.

First the start and stop frequencies are read using the OA (output active function) command. The start frequency is stored in r2 and the stop frequency is stored in r3; both are used in later subroutines.

The trace data is then read as a 1000 element array of vertical points, with the element number representing the X coordinate and the vertical magnitude representing the Y coordinate. The displayed magnitudes range from a minimum of 0 to a maximum of 1040 in screen coordinates. This creates a square field of points, each side of which is 1000 points in length, which leads to the Cartesian coordinate system explained above.

The FOR-NEXT loop contained in lines 113-116 is the portion of the program which transfers the raw data from the spectrum analyzer to the controller.

Line 115 is of particular interest. The first portion of the line divides each element of the raw data array ($L[I]$) by 1000 and multiplies by 10 to convert the screen magnitude to dB referenced to the lowest line on the screen graticle. The next portion of line 115 (R-100) establishes the reference level for the top line of the graticle (using subroutine REDRL) and translates that reference level to the bottom line of the screen graticle. The two numbers are then added, producing an absolute reading in dBuV.

At this point, the data stored in $L[I]$ should correlate with the raw data read on any EMI receiver, if the bandwidths are the same, the data-point frequencies are the same, and the instrument tolerances are accounted for.

```
102: "REDSA":  
103: "REDSA READS THE TRACE DATA FROM THE 8568A":  
104: "first read the start and stop frequencies":  
105: fmi  
106: wrt "sa","FA 0A"  
107: red "sa",T;T+r2  
108: wrt "sa","FR 0A"  
109: red "sa",T;T+r3  
110: wrt "sa","HD"  
111: "now read the data in screen (X,Y) coordinates":  
112: wrt "sa","01 TA"  
113: for I=1 to 1001  
114: red "sa",L[I]  
115: L[I]/1000*100+(R-100)+L[I]  
116: next I  
117: ret  
118:  
119:
```

N. CHOP

This subroutine compensates for the small (22.5 K byte) memory size of the HP9825. The 9825 operating system uses 8 bytes for each integer or real number stored. This means that the 1000 element array containing the raw data ($L[*]$) would occupy 8 K bytes (about 40%) of the available memory. Subroutine CHOP alleviates this problem. It chops the 1000 element $L[*]$ to a 100 element array, using the following algorithm:

- 1) The 1000 element array is divided into 100 sectors of 10 elements each.
- 2) The maximum value over the ten element sector is retained and the other points are discarded.

```
287: "chop":  
288: for I=11 to 1001 by 10  
289: 0->T  
290: for J=1 to 10  
291: max(T,L[I-J])>T  
292: next J  
293: T+L[(I-1)/10]  
294: next I  
295: ret  
296:
```

O. SMOOTH

This subroutine, called for broadband data only, rejects the narrowband data present in the broadband measurement. This is accomplished by limiting the changes between sampling points to 3 dB or less. Any changes greater than 3 dB are smoothed (or averaged) to a lower level. Lines 299 through 301 transfer the present trace data to another array for manipulation. Lines 302 through 305 calculate the changes in dB between each point. Lines 306 through 308 check to see if all delta values are less than 3 dB. If so, control is passed to line 317, and the smoothed data is placed back in array L[*]. If any delta value is 3 or greater, control is passed to the smoothing portion of the program (lines 310-315). The actual smoothing is done by lines 312 through 314, based on the delta values as follows.

If $K_I > 3$ and $K_{I+1} < -3$, a true spike is indicated. This spike is removed by replacing it with the average of the points on either side; that is:

$$Q[I] = \frac{K[I-1] + K[I+1]}{2}$$

If $K_I > 3$ only, then the following point $Q[I+1]$ is bigger than $Q[I]$. $Q[I+1]$ will then be replaced by the smaller value plus half the delta:

$$Q[I+1] = Q[I] + K[I]/2 .$$

If $K[I] < -3$, it indicates that $K[I+1]$ is much less than $K[I]$. In this case $Q[I]$ will be averaged down:

$$Q[I] = Q[I] + K[I]/2 .$$

After the averaging process, the program loops back to line 302, and continues to run until all delta values are less than 3 dB.

```
297: "smooth";
298: "rejects narrowband noise in broadband data";
299: for I=1 to 100
300: L[I]→Q[I]
301: next I
302: "calculate deltas";
303: for I=1 to 99
304: Q[I+1]-Q[I]+K[I]
305: next I
306: for I=1 to 99
307: if abs(K[I])>3;gto +3
308: next I
309: gto +8
310: for I=1 to 99
311: if I=1 or I=99;gto +2
312: if K[I]>3 and K[I+1]<-3;(Q[I-1]+Q[I+1])/2→Q[I];gto +3
313: if K[I]>3;Q[I]+K[I]/2→Q[I+1]
314: if K[I]<-3;Q[I]+K[I]/2→Q[I]
315: next I
316: gto "calculate deltas"
```

```
317: for I=1 to 100  
318: Q[I]→L[I]  
319: next I  
320: ret  
321:
```

P. BCOR

Subroutine BCOR, called for broadband measurements only, corrects the broadband data for the effects of the impulse bandwidth of the receiver. Lines 324-326 read the spectrum analyzer resolution bandwidth. Line 327 calculates the impulse bandwidth correction factor:

$$B = 20 \log \frac{RBW}{1.0E6}$$

Lines 328 through 330 add B to the data array.

```
322: "BCOR":  
323: "CORRECT FOR THE 856RA BANDWIDTH FACTORS":  
324: wrt "sa","RB DA"  
325: red "sa".T  
326: wrt "sa","HD"  
327: 20*log(1e6/T)→B  
328: for I=1 to 100  
329: L[I]+B→L[I]  
330: next I  
331: ret  
332:
```

Q. ANTCOR and ANTCORLIM

These subroutines are closely related and will be considered at the same time.

ANTCOR takes the data array output from CHOP or BCOR and corrects for the effects of the antenna factor. The corrected data is then plotted on the graph generated in GRAT. Due to the complexity of ANTCOR, it will be explained in detail. Line 123 calculates the frequency difference between sampling points by subtracting the start frequency (r_2) from the stop frequency (r_3) and dividing by the number of points. Line 127 calculates the frequency of each measurement point and converts to an integral number of megahertz. Line 128 calls ANTCORLIM which supplies the slope and intercept to generate a log-linear function used to approximate the frequency currently contained in r_1 . (The approximation will be explained in detail later). Line 130 initializes r_6 . Line 131 calculates the antenna correction factor. Line 133 adds the antenna factor to the data. Lines 135 and 136 positions the pen to write on the broadband or narrowband plot, based on the value of the X offset variable, Z. Lines 137 and 138 actually plot the corrected data (each datapoint of $L[I]$). Line 137 is necessary to plot the first point in each range; line 138 plots all others.

ANTCORLIM contains a set of variables used to generate a linear approximation to the antenna factor. These variables were calculated using the following procedure:

- 1) The antenna correction factors given by the manufacturer were plotted on semi-log paper.
- 2) The curve on the log chart was then approximated by a series of straight lines over frequency intervals.
- 3) The starting and stopping frequencies for each line were noted, and the slope and intercept of each linear segment were calculated using the following relationships:

$$S = \frac{r_4 - r_2}{\log(r_3) - \log(r_1)} \quad I = r_2 - S \log(r_1)$$

r_1 = interval starting frequency

r_2 = starting amplitude

r_3 = stop frequency

r_4 = stop amplitude

S = slope

I = Intercept

These slopes and intercepts then enable the calculation of the antenna factor for any datapoint if the frequency is known. The frequencies were calculated in line 127 using the semilog relationship:

$$Y = m \log x + b$$

```

120: "ANICUR";
121: "THIS ROUTINE CORRECTS FOR ANTENNA FACTORS";
122: "calculate the frequency delta";
123: (r3-r2)/100+r4
124: if A$="BB";pen# 2
125: if A$="NB";pen# 3
126: for I=1 to 100
127: (r2+(I-1)*r4)/1e6+r5
128: gsb "ANTCORI.IM"
129: "now generate the polynomial approximation to the antenna factor";
130: 0+r6
131: S*log(r5*1e6)+A+r6
132: "now correct the datapoint";
133: L[I]+r6+L[I]
134: "and plot it";
135: if A$="BD";0+Z
136: if A$="NB";1200+Z
137: if I=1;plt 2.0620210915e2*(log(r5*1e6)-4.146128036)+Z,L[I],-2;gto +2
138: plt 2.0620210915e2*(log(r5*1e6)-4.146128036)+Z,L[I]
139: next I
140: pen
141: ret
142:

```

```

333: "ANTCORLIM":
334: if r5>=1.4e-2 and r5<=2e-2;1.162025322e1+S;-8.497905766e1+A;ret
335: if r5>=2e-2 and r5<4e-2;1.328771238e1+S;-9.215084952e1+A;ret
336: if r5>=4e-2 and r5<6e-2;1.760450812e1+S;-1.120170025e2+A;ret
337: if r5>=6e-2 and r5<1e-1;3.245454497e1+S;-1.829727198e2+A;ret
338: if r5>=1e-1 and r5<1.5e-1;3.975211512e0+S;-4.057605756e1+A;ret
339: if r5>=1.5e-1 and r5<2e-1;7.2035305e1+S;-3.928613126e2+A;ret
340: if r5>=2e-1 and r5<4e-1;7.640434619e0+S;-5.150217309e1+A;ret
341: if r5>=4e-1 and r5<6e-1;0e0+S;-8.7e0+A;ret
342: if r5>=6e-1 and r5<8.5e-1;1.302911592e0+S;-1.572842025e1+A;ret
343: if r5>=8.5e-1 and r5<1e0;-1.133448319e1+S;6.030689912e1+A;ret
344: if r5>=1e0 and r5<1.6e0;1.371742229e1+S;-8.980453376e1+A;ret
345: if r5>=1.6e0 and r5<2e0;-1.031885116e1+S;5.931939065e1+A;ret
346: if r5>=2e0 and r5<4e0;2.491446071e1+S;-1.626867643e2+A;ret
347: if r5>=4e0 and r5<6e0;1.817239548e1+S;-1.181752431e2+A;ret
348: if r5>=6e0 and r5<8e0;-1.760863011e1+S;1.243539582e2+A;ret
349: if r5>=8e0 and r5<1e1;4.643483021e1+S;-3.177438115e2+A;ret
350: if r5>=1e1 and r5<1.5e1;6.814648305e0+S;-4.040253814e1+A;ret
351: if r5>=1.5e1 and r5<2e1;7.443648184e1+S;-5.256629867e2+A;ret
352: if r5>=2e1 and r5<3e1;-1.135774717e1+S;1.929232528e1+A;ret
353: if r5>=3e1 and r5<4e1;2.5625529e1+S;-1.807081637e2+A;ret
354: if r5>=4e1 and r5<5e1;-2.063770232e1+S;1.708890511e2+A;ret
355: if r5>=5e1 and r5<6e1;-3.409898347e1+S;2.745270509e2+A;ret
356: if r5>=6e1 and r5<7e1;-1.941840687e1+S;1.603393056e2+A;ret
357: if r5>=7e1 and r5<8e1;1.034626412e1+S;-7.316745634e1+A;ret
358: if r5>=8e1 and r5<9e1;1.954937807e1+S;-1.459004941e2+A;ret
359: if r5>=9e1 and r5<1e2;1.180134647e2+S;-9.291077175e2+A;ret
360: if r5>=1e2 and r5<1.1e2;-2.657474373e1+S;2.275979498e2+A;ret
361: if r5>=1.1e2 and r5<1.2e2;2.381673128e1+S;-1.776196887e2+A;ret
362: if r5>=1.2e2 and r5<1.3e2;-2.876695653e1+S;3.804134557e1+A;ret
363: if r5>=1.3e2 and r5<1.4e2;-6.835549619e1+S;5.693326239e2+A;ret
364: if r5>=1.4e2 and r5<1.5e2;6.007364352e1+S;-4.768675917e2+A;ret
365: if r5>=1.5e2 and r5<1.6e2;4.638099184e1+S;-3.649152219e2+A;ret
366: if r5>=1.6e2 and r5<1.7e2;6.456773725e1+S;-5.141214634e2+A;ret
367: if r5>=1.7e2 and r5<1.8e2;2.819899046e1+S;-2.147903506e2+A;ret
368: if r5>=1.8e2 and r5<1.9e2;-5.110494019e1+S;4.398852076e2+A;ret
369: if r5>=1.9e2 and r5<2e2;9.875924846e1+S;-8.008034838e2+A;ret
370: if r5>=2e2 and r5<3e2;-3.236957945e1+S;2.927008499e2+A;ret
371: if r5>=3e2 and r5<4e2;-2.401176834e0+S;3.865506718e1+A;ret
372: if r5>=4e2 and r5<5e2;2.063770232e1+S;-1.595267534e2+A;ret
373: if r5>=5e2 and r5<6e2;1.010340251e1+S;-6.788919537e1+A;ret
374: if r5>=6e2 and r5<7e2;2.539330128e1+S;-2.021062394e2+A;ret
375: if r5>=7e2 and r5<8e2;3.62119244e1+S;-2.977980217e2+A;ret
376: if r5>=8e2 and r5<9e2;9.774689033e0+S;-6.242493605e1+A;ret
377: if r5>=9e2 and r5<=1e3;8.741748127e0+S;-5.317564314e1+A;ret
378: ret
*23913

```

REFERENCES

1. Hewlett-Packard 9825A Calculator Operating and Programming, Hewlett-Packard Calculator Products Division, Loveland, Colorado, 1976.
2. Hewlett-Packard 9825A Calculator 9872A Plotter Programming, Hewlett-Packard Calculator Products Division, Loveland, Colorado, 1977.
3. Hewlett-Packard 9825A Calculator General I/O Programming, Hewlett-Packard Calculator Products Division, Loveland, Colorado, 1976.
4. Hewlett-Packard 9825A Calculator Extended I/O Programming, Hewlett-Packard Products Division, Loveland, Colorado, 1976.
5. Hewlett-Packard 9825A Calculator Matrix Programming, Hewlett-Packard Calculator Products Division, Loveland, Colorado, 1976.
6. Hewlett-Packard 9825A Calculator Advanced Programming, Hewlett-Packard Calculator Products Division, Loveland, Colorado, 1976.
7. Hewlett-Packard 9825A Calculator String Variable Programming, Hewlett-Packard Calculator Products Division, Loveland, Colorado, 1976.
8. 8568A Spectrum Analyzer Remote Operation, Hewlett-Packard Company, Santa Rosa, California, 1978.
9. 8568A Spectrum Analyzer Operation, Hewlett-Packard Company, Santa Rosa, California, 1977.
10. E.M.I. Measurement Procedure, Application Note 142, Hewlett-Packard Company, Palo Alto, California, 1972.
11. An Example of Automatic Measurement of Conducted EMI with the HP8568A Spectrum Analyzer, Application Note 270-1, Hewlett-Packard Company, 1978.
12. Electromagnetic Interference and Compatibility, Volume 2, Don White Consultants, Inc., 1976.
13. Electromagnetic Interference and Compatibility, Volume 6, Don White Consultants, Inc., 1975.

APPENDIX A. RESOLUTION AND VIDEO BANDWIDTH SETTINGS FOR EACH FREQUENCY BAND.

LOOP CTR	FREQUENCY BAND		BROAD BAND		NARROW BAND	
	START	STOP	RES BW	VID BW	RES BW	VID BW
1	14 KHz	100 KHz	1 KHz	3 MHz	100 Hz	100 Hz
2	100 KHz	1 MHz	10 KHz	3 MHz	300 Hz	300 Hz
3	1 MHz	10 MHz	100 KHz	3 MHz	3 KHz	3 KHz
4	10 MHz	20 MHz	100 KHz	3 MHz	3 KHz	3 KHz
5	20 MHz	200 MHz	1 MHz	3 MHz	3 KHz	3 KHz
6	200 MHz	1 GHz	1 MHz	3 MHz	10 KHz	10 KHz

APPENDIX B. PROGRAM TO CALCULATE SEMILOG SLOPES AND INTERCEPTS.

```
0: fnt
1: fnt 1,2x,e10,12x,e10,9x,e10,10x,e10
2: fnt 2,e16.12,5x,e16.12,5x,e16.12,5x,e16.12
3: wrt b.1,"START FREQ","STOP FREQ","SLOPE","INTERCEPT"
4: ent "START FREQUENCY IN HZ",r1
5: ent "AMPLITUDE AT START FREQ",r2
6: ent "STOP FREQUENCY (IN HZ",r3
7: ent "STOP AMPLITUDE",r4
8: "calculate the slope"
9: (r4-r2)/(log(r3)-log(r1))+S
10: "calculate the intercept"
11: r2-S*log(r1)+A
12: wrt b.2,r1,r3,S,A
13: r3+r1;r4+r2
14: qte b
#23412
```

APPENDIX C. LIST OF PROGRAM VARIABLE ASSIGNMENTS.

- A Intercept value used in linear approximation to logarithmic function (with slope value S).
- B Impulse bandwidth correction factor.
- D Temporary storage variable used in "grat."
- H Temporary storage variable used in "grat."
- I Variable used as for/next loop counter.
- J Variable used as for/next loop counter.
- R Reference level.
- T Temporary storage register, usually used to store information read from a bus instrument.
- S Slope value used in linear approximation to logarithmic function.
- Z Offset variable for plotting routines. Value is 0 for Broadband and 1200 for Narrowband measurements.

ARRAYS

- K[*] Point to point changes (delta array) produced in subroutine "SMOOTH."
- L[*] Primary data manipulation array used to contain data read from spectrum analyzer.
- Q[*] Temporary storage array used to contain data during manipulation in "SMOOTH."
- A\$ Broadband/Narrowband marker.
- D\$ String array used to contain run/test ID information in TITLE1 and title-out.
- T\$ Temporary storage area for string data.
- r1 Loop counter for the six frequency bands.
- r2 Stop frequency.
- r3 Start frequency.
- r4 Frequency change between points.
- r5 Frequency of each point
- r6 Calculated antenna factor for each point.

APPENDIX D. EQUIPMENT LIST.

<u>Item No.</u>	<u>Quantity</u>	<u>Description</u>
1	1	Spectrum Analyzer HP8568A
2	1	Calculator, HP9825A
3	1	Plotter, HP9872A
4	1	Printer, HP2631A (opt)
5	1	Printer, HP9866B (opt)
6	1	Antenna, Emco 3301
7	1	Antenna, Emco 3104
8	1	Antenna, Emco 3108
9	1	Disk, HP9885M (opt)




Figure 5. Instrumentation setup.

APPENDIX E. WHY DID WE . . . ?

In order to alleviate reader confusion and eliminate unnecessary correspondence, this appendix will answer some of the most frequently asked questions regarding specific programming steps contained in this program.

1. Q - Why did we specify in line 43 the attenuation to be used by the spectrum analyzer?

A - The HP8568A Spectrum Analyzer automatically adjusts the input attenuation to maintain an optimum output level from the first mixer. However, this causes a resultant change in the noise floor and shifting in the CRT trace level. Since the data trace is read as a matrix of y-position vs x-position values, this automatic attenuation adjustment would create inaccuracies in corrected data. Therefore, we just removed all attenuation by setting 0 dB attenuation.

2. Q - Why did we use output format 01 (y-position vs x-position) instead of format 03 (output in dB) in line 112?

A - A bug in the HP8568A Spectrum Analyzer firmware causes the HPIB to hang up if the 8568A outputs a 0 dB datapoint while in format 03, requiring program restart and consequent loss of data.


3. Q - Why do we use multiple scans instead of one long slow scan to capture broadband data?

A - A bug in the HP8568A firmware causes a noticeable frequency inaccuracy when using the coupled SWEEP TIME mode.

4. Q - Why are all broadband measurements taken using the 3 mHz video bandwidth setting?

A - In order to pass the transients associated with the broadband noise, it is necessary to have the video bandwidth as wide as possible.

APPENDIX F. TYPICAL TEST RESULTS



DISTRIBUTION

	No. of Copies
IIT Research Institute ATTN: GACIAC 10 West 35th Street Chicago, IL 60616	1
DRSMI-R	1
-RT, Mr. Black	1
-RTR, Mr. Daniels	1
Mr. Snead	5
-RPR	5
-RPT, Record Copy	1
Reference Copy	1

**DATE
ILME**